

## ÁLGEBRA DE CAMPOS E OBJETOS

**João Pedro Cordeiro**

**Cláudio Clemente Faria Barbosa**

**Gilberto Câmara**

### 8.1 INTRODUÇÃO

O que distingue um SIG de outros tipos de sistemas de informação são as funções que realizam análise espacial. Tais funções utilizam os atributos espaciais e não espaciais das entidades gráficas armazenadas na base de dados espaciais e buscam fazer simulações (modelos) sobre os fenômenos do mundo real, seus aspectos ou parâmetros.

A partir do modelo de dados apresentado no capítulo anterior foi estabelecida uma taxonomia para as diversas operações de análise geográfica, que serão divididas em: *operadores sobre objetos*, *operadores sobre campos*, *operadores de transformação entre campos e objetos* e *operadores mistos entre objetos e campos*. Esta análise permitirá obter um entendimento formal sobre a natureza das operações em Geoprocessamento.

A literatura especializada apresenta um grande número de artigos sistematizando as operações em um SIG, como Goodchild (1987), Maguire and Dangermond (1991) e Burrough e McDonnell(1998). O problema é usualmente abordado desde perspectivas distintas: operações de consulta sobre objetos (Egenhofer, 1994) e operações de manipulação sobre campos (Tomlin, 1990), sem unificar as duas visões.

O tema “álgebra de mapas” foi popularizado a partir dos livro “Geographic Information System and Cartographic Modeling” (Tomlin, 1990) (ver também Tomlin e Berry (1979); Tomlin (1983) e Berry (1987). Esta foi a primeira abordagem em que se buscou explorar de uma maneira formal as propriedades dos dados representados em SIG, usualmente representados por mapas. Ferramentas como o GRID (ArcInfo), o IDRISI e o IDL, foram essencialmente concebidas sob tal paradigma, no qual operações de modelagem são representadas por seqüências de operações primitivas descritas através de uma linguagem que procura respeitar as propriedades dos tipos de dados envolvidos.

Os elementos da álgebra de mapas descrita por Tomlin consistem de mapas que associam a cada local de uma dada área de estudo um valor quantitativo (escalar, ordinal, cardinal ou intervalar) ou qualitativo (nominal). Dito de outra forma, o modelo de dados adotado por Tomlin consiste desses tipos de dados, ficando o significado das operações a eles aplicadas ao encargo do modelador.

A caracterização adotada no modelo de dados **Spring** deriva-se da diferenciação explícita entre objetos e campos (ver Camara, 1994). Isto tem conseqüências importantes, já que permite compreender melhor a natureza de cada operador. Este trabalho procura traçar uma ponte entre as duas abordagens tradicionais de modelagem de dados geográficos, baseadas nos conceitos de objeto e campo. No modelo de dados **Spring**, o conceito de **Categoria** é introduzido afim de acrescentar semântica aos campos e objetos envolvidos em um processo de modelagem. Dados são assim categorizados segundo os modelos **Temático**, **Numérico**, **Imagem**, **Cadastral** e **Objeto**, que podem então ser especializados para assim constituir os tipos de dados envolvidos na construção de modelos.

O conceito de campo está implícito em grande parte das operações envolvendo dados locais representados em estruturas matriciais ou vetoriais. Outras operações, envolvendo, por exemplo, estatísticas sobre conjuntos de locais, tais como vizinhanças e zonas, podem ser fonte de informações para compor atributos de objetos, também, no sentido contrário, a espacialização de atributos de objetos permite a visualização e interação com outros campos. O modelo **Cadastral** exerce um papel importante nessa interação entre campos objetos, além de servir de suporte para extração de propriedades topológicas envolvendo objetos. Essas propriedades topológicas constituem o fundamento para operações de consulta espacial, fortemente baseadas em álgebra relacional.

O uso de representação no formato matricial para campos é sugestivo, entretanto nada impede que outros formatos (vetorial, árvore etc.) sejam também explorados. A linguagem LEGAL, por exemplo, prevê uma extensa classe de operações zonais envolvendo campos e objetos. As operações envolvendo objetos são fortemente baseadas em representações vetoriais (mapas cadastrais), embora recentes trabalhos (ver Camara et. al., 2005; Cordeiro et al., 2005) estão voltados para uma maior generalização no sentido prático do conceito de álgebra de mapas.

## 8.2 OPERAÇÕES QUE RESULTAM EM CAMPOS

Segundo a maneira de selecionar os valores associados a locais envolvidos na caracterização de cada local de uma área de estudo, as operações de uma álgebra de mapas podem ser classificadas em: **locais**, de **vizinhança** e **zonais** (Tomlin, 1990), como veremos nas próximas subseções neste capítulo.

### 8.2.1 OPERAÇÕES LOCAIS

Operações locais resultam em campos cujos valores locais são função de valores associados ao mesmo local por uma ou mais representações de outros campos. Alguns exemplos incluem:

- “Calcular” índices de vegetação com base em dados do modelo **Imagem**, gerando **Imagens** ou grades **Numéricas** como resultado, através de operações aritméticas aplicadas a cada “pixel” das imagens envolvidas.
- “Atribuir” valores de vulnerabilidade a unidades territoriais, com base em princípios de ecodinâmica (Tricart, ?, 19xx) através de uma operação aritmética simples envolvendo uma ponderação dos temas Geologia, Pedologia, Relevo, Vegetação e Clima.
- “Fatiar” uma grade **Numérica** de altimetria, agrupando locais pertencentes a faixas específicas de altitude, por exemplo: (0–300m), (300–500m) e (> 500m) segundo classes **Temáticas**: “planície”, “planalto” e “serrana”.

Algumas operações locais envolvendo campos são extensões naturais de *operações matemáticas* e resultam em campos **Numéricos** e **Imagens**. Também algumas *transformações* entre dados de diferentes modelos, tais como a Ponderação, Fatiamento, Reclassificação e Espacialização, podem ser usadas para gerar campos **Numéricos** e **Imagens**, além de **Temáticos**, à partir de outros campos de qualquer modelos, inclusive o de **Objetos** quando representados segundo o modelo **Cadastral**, como é o caso do Fatiamento e Reclassificação. Talvez a mais importante classe de operações locais, devido à sua importância na caracterização de critérios envolvendo diferentes campos e objetos, é a de *operações booleanas*, que também corresponde a uma extensão natural, que explora relações de comparação e operações lógicas aplicadas aos dados associados a cada local de uma área de estudo.

#### 8.2.1.1 OPERAÇÕES MATEMÁTICAS

Operações que resultam em grades **Numéricas** e **Imagens**; derivam das operações aritméticas, funções matemáticas, estatísticas, e muitas outras tipicamente aplicadas a dados de natureza quantitativa (cardinal, ordinal, intervalar etc.) dados por outras grades e imagens. A diferença entre as duas *categorias* é que imagens são

estruturas que envolvem apenas valores numéricos inteiros, enquanto as grades podem armazenar números reais em geral.

Para ilustrar, considere o conceito de “índice de vegetação” (ivdn), definido pela diferença normalizada entre valores radiométricos locais dados pelas bandas 3 (vermelho) e 4 (infravermelho próximo) de uma imagem multi-espectral, segundo a seguinte expressão algébrica:

$$\text{ivdn} = (\text{ivp} - v) / (\text{ivp} + v)$$

A operação acima resulta em valores locais no intervalo fechado de números reais  $[-1, 1]$ , de modo que uma grade numérica deveria ser usada em sua representação. Entretanto, a fim de permitir a visualização como imagem, como ilustra a figura 5.1, o cálculo acima pode ser corrigido por parâmetros que garantam melhor distribuição dos resultados (inteiros) locais no intervalo  $[0, 255]$ . Assim, as regiões mais claras da imagem associada à variável “ivdn” correspondem às regiões mais claras, de índices mais altos efetivamente obtidos através da operação:

$$\text{ivdn} = (1 + (\text{ivp} - v) / (\text{ivp} + v)) * 255$$




Figura 5.1 – Cálculo de índices de vegetação usando aritmética local

Um programa em LEGAL, que descreve essa operação poderia ser descrito como abaixo:

```
{
  Imagem v, ivp, ivdn  ("Imagem_TM");
  v = Recuperar (Nome = "TM_3");
  ivp = Recuperar (Nome = "TM_4");
  ivdn = Novo (Nome = "IndVerde", ResX=30, ResY=30 );
  ivdn = ( (ivp - v) / (ivp + v) + 1) / 2 * 255 ;
}
```

### 8.2.1.2 OPERAÇÕES DE TRANSFORMAÇÃO

São operações que mapeiam entre os diferentes categorias do modelo de dados Spring, tais como a *Ponderação*, que associa classes Temáticas a valores Numéricos; a *Reclassificação*, que define a associação de entre diferentes conjuntos de classes Temáticas, e o *Fatiamento*, que mapeia valores Numéricos locais a classes de alguma

categoria Temática. Atributos quantitativos ou qualitativos, de dados do modelo Objetos podem também ser usados como argumentos de transformações, em particular a operação de Espacialização permite o mapeamento entre atributos de objetos associados a mapas cadastrais e campos dos diversos modelos. Transformações são implementadas através do uso de *tabelas* específicas que descrevem o mapeamento desejado.

### PONDERAÇÃO

Uma operação de *Ponderação* consiste em obter um campo **Numérico** a partir de um campo **Temático**, de tal modo que cada local de uma área de estudo fique associado a um valor indicando o peso de cada classe temática diante de uma operação quantitativa que se deseje modelar. Por exemplo, a Figura 5.2 mostra uma operação de ponderação (conversão de um mapa de solos em um mapa de solos ponderado). Em que um conjunto de classes de solo { Latossolo, Podzólico, Litossolo, Hidromorfico}, é mapeado em valores reais segundo a tabela de mapeamento:

<b>Latossolo</b>	<b>0.9</b>
<b>Podzólico</b>	<b>0.8</b>
<b>Litossolo</b>	<b>0.7</b>
<b>Hidromorfico</b>	<b>0.4</b>




Figura 5.2 - Exemplo de operação de ponderação.

Um programa em LEGAL, que descreve essa operação poderia ser descrito como abaixo:

```
{
Tematico      solo_T ("Solos");
Numerico      solo_P ("Ponderacoes");
Tabela       pesos (Ponderacao);
```

```

solo = Recupere (Nome="Mapa_Solos");
pesos = Novo (Categoria = "Solos",
              "Latossolo"      : 0.9,
              "Podzolico"      : 0.7,
              "Litossolo"      : 0.6,
              "Hidromorfico"   : 0.4 );
solo_P = Novo (Nome="Solo_Ponderado", ResX=30, ResY=30,
              Escala=50000, Min = 0, Max = 1);
solo_P = Pondere (solo, pesos);
}

```

### FATIAMENTO EM CLASSES

Uma operação de *Fatiamento* consiste em obter um campo **Temático** a partir de um campo **Numérico**, de tal modo que cada local de uma área de estudo fique associado a um valor indicando, sob a forma de um conjunto de classes temáticas, os intervalos de valores registrados à partir de uma grade **Numérica**. Por exemplo, a figura 5.3 mostra um exemplo de uma operação de *Fatiamento* em um conjunto de classes {baixa, media, alta}, onde um mapa de declividade em graus é convertido para um mapa de classes de declividade segundo a tabela:

< 9	baixa
< 19	media
> 19	alta




Figura 5.3 - Exemplo de operação de fatiamento em classes.

Um programa em LEGAL, que descreve essa operação poderia ser descrito como abaixo:

```

{
Numerico alt      ("Altimetria");
Tematico altfat   ("Faixas_Altimetricas");
Tabela  faixas    (Fatiamento);

alt = Recupere (Nome = "Mapa Altimetrico");
faixas = Novo (CategoriaFim = "Faixas_Altimetricas",

```

```

[980,1020] : "BAIXA",
[1020,1100] : "MEDIA",
[1100,1180] : "ALTA" );
altfat = Novo (Nome = "ALtimetria_F", ResX=30, ResY=30,
Escala=50000);

altfat = Fatie (alt, faixas);
}

```

Uma aplicação comum de *Fatiamento* em classes consiste em obter um campo **Temático** a partir de um campo **Numérico** contendo as distâncias de cada ponto de uma área de estudo, a um objeto de referência (representado por um ponto, linha ou região). Trata-se de operação puramente geométrica (espacial). A Figura 5.4 ilustra esta operação.




Figura 5.4 - Exemplo de mapa de distâncias.

Com base em uma tabela que descreve o mapeamento entre faixas de distancia no campo **Numérico** e classes **Temáticas** de um mapa resultante (Classes\_de\_distancia), como descrito pelo trecho de programa abaixo:

```

Tabela Faixas (Fatiamento);
Faixas=Novo([0, 39]      : "perto",
            [30, 100]    : "proximo",
            [100, 500]   : "longe" );
Classes_de_distancia = Fatie (distancias, faixas)

```

## RECLASSIFICAÇÃO

Uma operação de *Reclassificação* consiste em obter um campo **Temático** a partir de outro campo **Temático**, que podem ou não, ser de categorias Temáticas distintas. Cada local de uma área de estudo é associado a um valor de um conjunto de classes temáticas, segundo uma tabela que modela o mapeamento entre os conjuntos de entrada e saída.

Figura 5.5 ilustra uma operação de *Reclassificação* de um mapa **Temático** de vegetação agrupando os conjuntos de classes de cobertura florestal {"caatinga",

“manguezal”, “restinga”, “agricultura”, “contato”}, gerando um novo mapa com as classes **Temáticas** do conjunto {"Alta", "Baixa" e "MuitoBaixa"}. segundo a tabela:

Caatinga	Alta
Manguezal	
Restinga	
Agricultura	Baixa
Contato	MuitoBaixa



Figura 5.5 - Exemplo de operação de Reclassificação

Com base nessa tabela que descreve o mapeamento entre dois conjuntos de classes **Temáticas**, usando a linguagem LEGAL, a operação de Reclassificação acima pode ser descrita por:

```
{
  Tematico veg ("Vegetacao");
  Tematico vegR ("Adequacao");

  veg = Recuperar (Nome="VegetacaoSudene");
  vegR = Novo (Nome="Veg_Reclass", ResX=1000, ResY=1000, Escala
= 2000000);
  Tabela reclass (Reclassificacao);
  reclass = Novo (CategoriaIni= "Vegetacao", CategoriaFim=
"Adequacao",
    "Caatinga", "Manguezal", "Restinga" : "Alta",
    "Agricultura" : "Baixa",
    "ContatoVegetacional" : "MuitoBaixa" );
  vegR= Reclassifique (veg, reclass);
}
```

## ESPACIALIZAÇÃO



Atributos de objetos podem ser usados como base para a geração de campos afim de representar a sua variação espacial. A operação *Espacilize* em LEGAL gera um campo **Numérico** ou **Temático** com a partir dos valores de algum atributo de objetos associados a elementos vetoriais de um mapa **Cadastral**.

A figura 5.6 ilustra o caso em que um mapa **Cadastral**, associado a **Objetos** que representam unidades de paisagem, é usado para a geração de um mapa **Temático** de solos com base no atributo “tipo de solo” de cada unidade. Aqui, os atributos indicam diferentes classificações do meio físico (geomorfologia, solos, geologia e vegetação).




Figura 5.6 - Exemplo da operação de espacialização.

Com base em variáveis que representam mapas cadastrais e objetos neles representados a operação de *Espacialização* acima pode ser descrita pelo programa:

```
{
Cadastral      cad ("Paisagem");
Objeto         obj ("Unidades");
Temático       tem ("Unidades de Paisagem");
cad = Recupere (Nome = "cadastro de ups");
tem = Novo(Nome="mapa de ups", ResX=5000, ResY=5000,
Escala=100000);
tem = Espacialize (obj."SOLO" OnMap cad);
}
```

Uma operação de *Espacialização* pode também resultar em dados de categoria **Númerica**. A figura 5.7 ilustra a espacialização do atributo “densidade populacional” de objetos Municípios sob a forma uma grade (campo) numérica.




Figura 5.7 - Exemplo de espacialização combinada com fatiamento.

O resultado acima na verdade combina uma operação de *Espacialização* e uma de *Fatiamento*, afim de gerar também como resultado, um mapa **Temático** de “faixas de densidade” populacional para uma certa área de estudo, como ilustra o programa abaixo:

```
{
Numerico      num ("Densidades");
Cadastral     cad ("Divisao Municipal");
Objeto        obj ("Municipios");
Tematico      tem ("Populações");
Tabela        tab (Fatiamento);

cad = Recupere (Nome="Municipios");
tab = Novo (CategoriaFim="Populações",
            [0, 100000] : "Baixa",
            [100000, 1000000] : "Media",
            [1000000, 10000000] : "Alta");
num = Novo (Nome="Densidade", ResX=5000, ResY=5000,
Escala=100000, Min=0.5, Max=7100.0);
tem = Novo(Nome="Faixas de Densidade", ResX=5000, ResY=5000,
Escala=100000);

num = Espacialize (obj."POPULACAO" NoMapa cad);
tem = Fatie (num, tab);
}
```

Além da *Espacialização*, também as operações baseadas em tabelas de mapeamentos, como a *Reclassificação*, a *Ponderação* e o *Fatiamento*, podem fazer uso de algum atributo adequado de **Objetos**. Numa operação de *Fatiamento* por atributo, por exemplo, um campo temático é gerado a partir de algum atributo de tipo “texto” cujo conteúdo possa ser entendido como classes temáticas. A operação dupla ilustrada no exemplo anterior poderia ser descrita alternativamente como uma única,

de *Fatiamento* aplicado diretamente sobre os valores de atributos de objetos, sem a necessidade de gerar uma grade numérica intermediária para isso, como ilustrado pela figura 5.8 e pelo programa em LEGAL descrito em seguida.




Figura 5.8 – Fatiamento por Atributos

```
{
Cadastral      cad ("Divisao Municipal");
Objeto         obj ("Municipios");
Tematico       tem ("Populações");
Tabela         tab (Fatiamento);

tab = Novo (CategoriaFim="Densidade Populacional",
           [0, 100000] : "Baixa",
           [100000, 1000000] : "Media",
           [1000000, 10000000] : "Alta");
cad = Recupere (Nome = "Mapa de Municipios");
tem = Novo (Nome="Faixas de Densidade", ResX=5000, ResY=5000,
           Escala=100000, Repres=Vetor);

tem = Fatie (obj."POPULACAO" NoMapa cad, tab);
}
```

### 8.2.1.3 COMPARAÇÕES E OPERAÇÕES BOOLEANAS

Muitas implementações de álgebra de mapas pressupõem a existência de um modelo de dados adicional que pode ser chamado **Booleano**, por sua natureza binária. Resultados de operações **Booleanas** são os campos obtidos à partir da comparação entre valores locais de outros campos quantitativos ou qualitativos, baseada em relações de ordem ou igualdade. Além de comparações toda a riqueza da álgebra Booleana baseada em operações primitivas como E, OU e NÃO também pode ser utilizada afim de caracterizar as mais diversas situações que podem ocorrer em locais de uma área de estudo, com base em grades, imagens e mapas temáticos disponíveis em um certa base de dados.

Uma operação do tipo  $A \text{ E } B$ , onde A e B representam campos **Booleanos** permite selecionar todos os elementos da interseção entre A e B;  $A \text{ MENOS } B$  retorna elementos contidos exclusivamente em A;  $A \text{ OU } B$  retorna todos os elementos contidos tanto em A como em B;  $A \text{ XOU } B$  retorna todos os elementos contidos em A e B não incluídos na interseção de A e B.

A linguagem LEGAL permite descrever operações locais de comparação e Booleanas, mas não oferece um modelo explícito **Booleano** para representar campos que poderiam resultar de tais operações. Na verdade tais operações são utilizadas como argumentos de outras operações que podem ser **Temáticas**, **Numéricas** e de **Imagens**. Sob o ponto de vista da linguagem expressões Booleanas são usadas para descrever regiões usadas como argumentos em outras operações.

### ATRIBUIÇÃO CONDICIONAL

Operações Booleanas são usadas junto a operações de atribuição condicional, permitindo assim a caracterização de locais segundo operações alternativas. O critério de escolha entre uma alternativa ou outra resulta da avaliação de uma operação **Booleana**. Segue-se um exemplo descrito em LEGAL:

```
indices =   altimetria < 600 AND cobertura == "floresta" ?  
           (ivp - v) / (ivp + v) :  
           (ivp - a * v - b) / sqrt (sqr (a) + 1) ;
```

Aqui variáveis representam uma grade **Numérica** de altitudes, bandas de uma **Imagem** multi-espectral e um mapa **Temático** de cobertura vegetal. Operações alternativas para o cálculo de índices de vegetação são usadas dependendo do resultado da avaliação de uma expressão algébrica Booleana. A descrição de uma operação como acima segue a forma sintática:

```
<resultado> = <booleana> ? <caso sim> : <caso não> ;
```

Um outro de exemplo de atribuição condicional é dado pelo operador *Atribua*. Aqui, classes temáticas podem ser explicitamente associadas a locais de uma área de estudo, dependendo do resultado da avaliação de diferentes condições **Booleanas** envolvendo campos de diferentes naturezas. É portanto uma operação **Temática**, que pode ser descrita em LEGAL segundo a forma abaixo:

```
<resultado> = Atribua {  
    "classe1" : <booleana>,  
    "classe2" : <booleana>,  
    ...  
    "classeN" : <booleana> } ;
```

Como exemplo de uso de uma operação *Atribua*, tome-se o caso de determinar um mapa de aptidão agrícola a partir de um banco de dados contendo mapas de solo, declividade, precipitação, junto ao conjunto de regras expresso na tabela abaixo e ilustrado na figura 5.9.

Aptidão	Precipitação	Capacidade
Alta	> 100 mm	0-3.5%
Média	100-50 mm	3.5-12%
Baixa	< 50 mm	>12%




Figura 5.9 – Atribuição condicional

Usando a linguagem LEGAL, tal resultado poderia ser expresso da seguinte maneira:

```
{
Numerico capacidade ("Solos");
Numerico precipitacao ("Precipitacao");
Tematico adequacao ("Adequacao");

capacidade = Recuperar (Nome="Capacidade de Agua");
precipitacao = Recuperar (Nome="Precipitação média");
adequacao = Novo (Nome="Adequacao", ResX=10000, ResY=10000,
Escala=2000000);

adequacao = Atribua {
  "Alta" :
    precipitacao > 6 &&
    capacidade > 150,
  "Media" :
    precipitacao > 3 &&
    capacidade >100.0,
  "Baixa" :
    precipitacao > 0 &&
    capacidade > 40.0
} ;
}
```

O apelo da abordagem *Booleana* é a sua simplicidade e fácil aplicação, com a combinação lógica de mapas em um SIG diretamente análoga ao tradicional método de sobreposição empregado nas mesas de luz. Porém, na prática pode não ser apropriado atribuir igual importância para cada um dos critérios combinados, as evidências precisam ser pesadas dependendo da sua relativa importância (Bonham-Carter, 1994).

### 8.2.2 OPERAÇÕES DE VIZINHANÇA

Nesta classe de operações, dado um campo, computa-se o campo de saída com base na dimensão e forma de uma vizinhança em torno de cada local  $p$ . Exemplos incluem:

- *cálculos de estatísticas* como *mínimo*, *máximo*, *média*, *moda* para uma vizinhança em torno de um ponto;
- *filtros* para processamento de **Imagens**;
- *interpolação espacial* para grades **Numéricas** (por média, vizinho mais próximo etc);
- *mapas de declividade e exposição* para MNT;
- *mapas de índices de diversidade Temática* (cada local está associada ao número de diferentes classes associadas a seus vizinhos).

Como exemplo de operação de vizinhança, tome-se o caso da estimativa da diversidade de vegetação de uma região, computado a partir de uma vizinhança 3 x 3 em torno de cada ponto. A idéia é que a diversidade seja maior em áreas de contato ecológico entre regiões homogêneas. A Figura 5.10 apresenta um mapa de vegetação e mostra a estimativa de diversidade computada para uma parte do mapa.




Figura 5.10 - Exemplo de operação de “estimativa de diversidade”.

A linguagem LEGAL oferece alguns recursos para manipulação de elementos vizinhos na caracterização de locais de uma área de estudo, que consiste de referenciá-los explicitamente nas expressões, em termos do posicionamento relativo a cada local caracterizado (local focal). Veja por exemplo a expressão:

```
media = ( img[-1,-1] + img[-1, 0] + img[-1, 1]
          + img[ 0,-1] + img[ 0, 0] + img[ 0, 1]
          + img[ 1,-1] + img[ 1, 0] + img[ 1, 1] ) / 9
```

Aqui, a media dos valores associados a locais vizinhos, vindos de um campo **Imagem** representado pela variável `img`. Os locais vizinhos são referidos pelo deslocamento em termos de linhas (acima e abaixo) e colunas (à esquerda e à direita), relativamente ao foco associado ao par ordenado `[0,0]`. A seleção dos locais em uma vizinhança é uma questão de especificação adequada desse posicionamento relativo a cada local tomado como foco.

Algumas outras operações envolvendo vizinhanças em LEGAL são dadas mais implicitamente. É o caso de uma operação que resulta em um campo de resolução mais grossa que a dos campos tomados com argumentos (dados de entrada), pode-se usar critérios baseados em estatísticas simples, tais como “média”, “maioria”, “máximo” etc, entre valores locais de resolução mais fina interceptados, como ilustra a figura 5.11.



Figura 5.11 - Exemplo de operação de “generalização”.

No exemplo acima um campo representando a vegetação em uma área de estudo com resolução 10x10 é usado para determinar dois novos campos: no primeiro, também com resolução 10x10, cada local é associado à classe mais freqüente entre as classes associadas aos 8 locais vizinhos adjacentes no mapa original de vegetação; o segundo campo corresponde a um campo com resolução 30x30, tal que cada local é associado à classe predominante entre as classes

associadas aos 9 locais a que corresponde na imagem original. As expressões abaixo descrevem essa situação usando a linguagem LEGAL:

```
filtro_moda = (Maioria) original;  
generalização = (Maioria) original;
```

Operações como acima muitas vezes são referidas como operações de interpolação. Em LEGAL, sempre que diferentes resoluções estiverem envolvidas em uma operação, algum tipo de interpolação é aplicada, mesmo que implicitamente.

### 8.2.3 OPERAÇÕES ZONAIS

As operações *zonais* são definidas sobre regiões específicas de um campo de entrada, onde as restrições são fornecidas por zonas definidas através de operações **Booleanas** envolvendo outros campos, ou de polígonos, linhas e pontos, associados a **Objetos** através de mapas **Cadastrais**. Por exemplo, com base em um mapa de solos e um mapa de declividade da mesma região, pode-se obter a declividade média para cada tipo de solo; ou ainda determinar o percentual de cobertura florestal por municípios do estado do Amazonas, com base em um mapa de cobertura vegetal e em um mapa **Cadastral** de divisão municipal, associado a **Objetos** municípios. Nesta seção, apresenta-se o caso de operações zonais onde a restrição é definida a partir de um ou mais campos, como no primeiro exemplo acima. O segundo caso será discutido na seção “Operações Mistas”.

Os operadores zonais incluem as estatísticas simples tais como: *media, moda, máximo e mínimo, diversidade, desvio padrão* etc, envolvendo valores associados a locais de regiões específicas, dadas por alguma forma de restrição. Por exemplo, uma operação **MáximoZonal**, na qual se toma um conjunto de classes **Temática** de solo como restrição sobre uma grade **Numérica**, irá retornar o maior valor da grade para cada tema (ver Figura 5.12).






Figura 5.12 - Operação de máximo zonal.

Operações zonais permitem materializar num SIG conceitos, tais como o de unidade de paisagem (Bertrand, 1981; Tricart, 1977; Tricart e KiewietdeJonge, 1992) e área-unidade (Hartshorne, 1978) que permitem a ligação entre dados físico-bióticos e sócio-econômicos, essencial para estudos de ordenação do território.

Alternativamente aos procedimentos manuais e de forma análoga ao exemplo anterior, uma operação de *MediaZonal* pode ser conveniente na determinação da vulnerabilidade natural à erosão (Crepani et al, 1996) ou da fragilidade (Ross, 1990) das “unidades de paisagem” ou das “unidades territoriais básicas” mapeadas a partir das imagens de satélites. Toma-se o campo temático ou cadastral “unidade de paisagem” como restrição sobre campos numéricos (MNT), que por sua vez, são resultantes de ponderações sobre os campos temáticos de geologia, geomorfologia, solos, vegetação e clima. Obtém-se com o resultado um campo numérico, cujo valor médio para cada “unidade de paisagem” mapeada representa sua vulnerabilidade ou fragilidade natural. Este campo numérico (MNT) pode ser “fatiado” em classes de vulnerabilidade (muito alta, alta, média, baixa e muito baixa), obtendo-se um campo temático.

#### ATUALIZAÇÃO DE ATRIBUTOS DE OBJETOS A PARTIR DE CAMPOS

Uma das operações muito úteis e necessárias diz respeito à atualização de atributos de objetos a partir de dados de um campo. Por exemplo, considere a situação aonde temos:

- Uma mapa de municípios de uma região (cadastral).
- Um conjunto de municípios representados nesse mapa (objetos).
- Um mapa de altimetria da região (campo).

Neste caso, pode-se determinar, por exemplo, a altitude média por município e atualizar essa informação no banco de dados, criando um novo atributo para os objetos municípios, como ilustrado na Figura 5.13.




Figura 5.13 - Operação de Atualização de Atributos de Objetos.

Em LEGAL uma tal operação pode ser descrita como uma operação zonal, com base em variáveis que representam uma grade de altitudes (altitude), um mapa cadastral (municípios) e seus objetos associados (município). O atributo considerado não precisa estar associado a uma variável, ele pode ser dado explicitamente por seu nome entre aspas ("Altitude"). Veja a o programa abaixo:

```
{
Cadastral cad ("Cadastro_Urbano");
Objeto obj ("Quadras");
Numerico num ("Grade_Declividade");

cad = Recupere (Nome = "Quadras_New");
num = Recupere (Nome = "declividade");

obj."DECLIVID" = Media (num, obj OnMap cad);
}
```

### 8.3 RELACIONAMENTOS ESPACIAIS ENTRE OBJETOS

Em nosso modelo, os objetos estão sempre associados a representações gráficas 2D (*pontos, linhas e regiões*). Como as operações da álgebra de objetos podem envolver restrições espaciais, será fundamental caracterizar os relacionamentos espaciais, que podem ser divididos em (Güting, 1994):

- *relacionamentos topológicos*, tais como “dentro de” e “adjacente a”, invariantes a transformações biunívocas e bicontínuas (como as de escala, translação e rotação);
- *relacionamentos métricos*, derivados das operações de *distância* e *direção*. O cálculo destas operações pressupõe sempre a existência de um espaço métrico, o que pode não ser sempre o caso.

A definição de um conjunto mínimo de operadores é objeto de muito debate na literatura:

- Freeman (1975) define um conjunto de 13 operadores: “à esquerda de”, “à direita de”, “acima” (mais alto que, sobre), “abaixo” (sob), “atrás”, “próximo a”, “longe de”, “ao lado de” (adjacente a), “tocando”, “dentro de”, “fora de”, “entre”;
- Egenhofer (1994) usa “disjunto”, “encontram”, “igual”, “dentro de”, “contém”, “cobre”, “coberto por” e “sobreposição”.

Devido à variedade de propostas, o esforço para formalizar os relacionamentos espaciais é parte fundamental da proposta de uma álgebra de objetos para um conjunto bem-definido de objetos geográficos (casos “simples” de regiões sem buracos e linhas contínuas).

#### 8.3.1 ANÁLISE DE RELACIONAMENTOS TOPOLÓGICOS

Apresenta-se nesta seção uma análise dos relacionamentos topológicos entre elementos do tipo ponto-linha-área. Utilizam-se os termos propostos por Clementini et al. (1993) que, a partir da análise das configurações possíveis entre os elementos ponto-linha-área, propõem cinco nomes para os relacionamentos topológicos: *toca*, *dentro de*, *cruza*, *sobrepõe* e *disjunto*. Diz-se que um conjunto de pontos  $\omega_1$  *toca* outro conjunto  $\omega_2$  se a única coisa em comum entre eles está contida na união de suas fronteiras, como ilustram os exemplos da Figura 5.14.




Figura 5.14 - Exemplos de situações topológicas que ilustram o relacionamento *toca*, no caso de duas áreas (a, b), duas linhas (c, d), linha e área (e, f, g), um ponto e uma linha (h) e um ponto e uma área (i). Adaptado de Clementini et al. (1993).

Diz-se que um conjunto de pontos  $\omega_1$  *está dentro de* outro conjunto  $\omega_2$  quando a intersecção dos dois conjuntos de pontos é o próprio  $\omega_1$  (ver Figura 5.15).




Figura 5.15 - Exemplos do relacionamento “dentro de” (contido em).

Duas linhas se *cruzam* se sua intersecção ocorre em um ponto interno de ambas (note-se que a intersecção de seus pontos limites será definida como *toca*); de forma similar, uma linha cruza uma área se o interior da linha está parcialmente interno e parcialmente externo a esta área. Diz-se ainda que dois conjuntos de pontos  $\omega_1$  e  $\omega_2$  possuem *sobreposição* quando o resultado de sua intersecção é uma figura da mesma dimensão de ambos. Este relacionamento é aplicável apenas a casos de elementos homogêneos. Finalmente, duas figuras são *disjuntas* se sua intersecção é vazia. Vejam-se os exemplos da Figura 5.16.




Figura 5.16 - Exemplos de relacionamentos:

1. *Cruza* entre duas linhas (a), linha e área (b, c).
  2. *Sobreposição* entre duas áreas (d), duas linhas (e, f).
  3. *Disjunto* entre duas áreas (g), linha e área (h), dois pontos (i).
- (Adaptado de Clementini et al. (1993))

#### 8.4 OPERAÇÕES SOBRE OBJETOS

Pode-se definir as seguintes operações sobre objetos:

1. *restrições sobre atributos*: computados em função dos atributos de entidades espaciais (e.g. “selecione todas as cidades de Alagoas com mortalidade infantil maior que 10%”);
2. *restrições espaciais*: derivados a partir dos relacionamentos *topológicos* das entidades geográficas (e.g. “dê-me todas as escolas municipais do bairro Jardim Satélite”), de *direção* (“ao norte de”, “acima de”) ou *métricos* (e.g. “dê-me todas as escolas a menos de 500 m da Via Dutra”);
3. *propriedades de objetos*: os resultados correspondem a predicados de um objeto ou de um conjunto de objetos (e.g. “calcule a média do valor venal das casas do bairro Jardim Esplanada” ou “indique o caminho ótimo para o ônibus que vai do Centro ao Jardim Uirá”).

Estas operações utilizam as primitivas definidas anteriormente: as relações topológicas *toca*, *dentro de*, *disjunto*, *cruza* e *sobreposição*, as relações métricas unárias (comprimento, área, perímetro) e binárias (distância, direção).

#### 8.4.1 SELEÇÃO POR ATRIBUTOS

O operador de seleção por atributos sobre um conjunto de objetos  $GO$ , dada uma restrição baseada apenas nos atributos descritivos de  $GO$ , gera como resultado um sub-conjunto  $GO' \subset GO$ , cujos membros satisfazem a restrição.

Esta é uma operação semelhante à seleção da álgebra relacional, como indica o exemplo: "Recupere as cidades do Estado de São Paulo com população entre 100.000 e 500.000 habitantes".

#### 8.4.2 SELEÇÃO ESPACIAL

Para definir as operações de consulta espacial, é necessário lançar mão do conceito de predicado espacial. Dados uma região geográfica  $R$ , um conjunto de objetos  $GO$  e um mapa cadastral que mapeia objetos de  $GO$  em  $R$ , um *predicado espacial* é uma restrição espacial definida através de um relacionamento topológico (*dentro de*, *toca*, *cruza*, *subposição* e *disjunto*) ou de um relacionamento métrico (*distância*).

Intuitivamente, os predicados espaciais utilizados nas operações envolvendo objetos são assertivas do tipo “rio que cruza o município de São José dos Campos, no mapa do Vale do Paraíba”.

Assim, dados uma região geográfica  $R$ , um conjunto de objetos  $GO$ , um mapa cadastral que mapeia objetos de  $GO$  numa região geográfica  $R$  e um *predicado espacial*, o operador de *seleção espacial* é tal que o resultado desta operação é um subconjunto do conjunto original composto de todos os objetos que satisfazem o predicado espacial, como ilustrado no exemplo da Figura 5.17:

- “selecione todas as regiões da França adjacentes à região de Midi-Pirenées (que contém a cidade de Toulouse)”.




Figura 5.17 - Exemplo de operação de seleção espacial.

#### 8.4.3 JUNÇÃO ESPACIAL

O termo *junção espacial* é empregado por analogia à operação de *junção* em banco de dados convencionais e denota o conjunto de operações onde ocorre a comparação entre dois conjuntos de objetos, baseado num predicado espacial computado sobre suas representações. A operação de junção espacial é tal que produz como resultado uma *coleção de objetos e valores* que satisfazem a restrição espacial. Exemplos:

- “para cada estrada da Amazônia, ache as reservas indígenas a menos de 5 km de uma estrada”;
- “para as cidades do sertão cearense, ache quais estão a menos de 10 km de algum açude com capacidade de mais de 50.000 m<sup>3</sup> de água”;

No primeiro exemplo a resposta é um conjunto de pares (reserva, estrada); no segundo, um conjunto de pares (cidade, açude).

## BIBLIOGRAFIA

- ASSAD, E.D.; SANO, E.E. *Sistemas de Informações Geográficas - Aplicações na Agricultura*. Brasília, EMBRAPA, 1998 (2<sup>a</sup>. edição).
- BARBOSA, C.C.F.; CAMARA, G.; MEDEIROS, J.S.; CREPANI, E.; NOVO, E.M.L.M. ; CORDEIRO, J.P.C. “Operadores Zonais em Álgebra de Mapas e Sua Aplicação a Zoneamento Ecológico-Econômico”. In: Simpósio Brasileiro de Sensoriamento Remoto, Salvador, Bahia, 1998. Anais, INPE, 1998 (em CD-ROM).
- BERTRAND, G. *Paisagem e Geografia Física Global: Esboço metodológico*. Caderno de Ciências da Terra, 13. Inst. de Geografia-USP. São Paulo. 1971.
- BONHAM-CARTER, G.F. - *Geographic Information Systems for Geoscientists: Modelling with GIS*. Pergamon, Oxford, 1994.
- BURROUGH, P. A.; MCDONNELL, R.A. *Principles of geographical information systems*. Oxford, Oxford University Press, 1998.
- CÂMARA, G. *Modelos, Linguagens e Arquiteturas para Bancos de Dados Geográficos*. Tese de Doutorado em Computação Aplicada. São José dos Campos, INPE, 1995. [www.dpi.inpe.br/teses/gilberto](http://www.dpi.inpe.br/teses/gilberto).
- CLEMENTINI, E.; DI FELICE, P.; VAN OOSTEROM, P. “A Small Set of Formal Topological Relationships Suitable for End-User Interaction”. In: Third International Symposium on Spatial Data Handling. Proceedings, Singapore, 1993, pp. 277-295.
- CORDEIRO, J. P.; AMARAL, S.; FREITAS, U.M.; CÂMARA, G. “Álgebra de campos e suas aplicações”. In: VIII Simpósio Brasileiro de Sensoriamento Remoto. Salvador. 14-19 de abril. 1996.
- CREPANI, E.; MEDEIROS, J. S.; HERNANDEZ FILHO, P.; FLORENZANO, T. G. DUARTE, V. AZEVEDO, L. G.. “Uso de sensoriamento remoto no zoneamento ecológico-econômico”. In: VIII Simpósio Brasileiro de Sensoriamento Remoto. Salvador. 14-19 de abril. 1996.
- EGENHOFER, M. “Spatial SQL: A Query and Presentation Language”. *IEEE Transactions on Knowledge and Data Engineering*, 6:86-95, 1994.
- FREEMAN, J. “The Modelling of Spatial Relations”. *Computer Graphics and Image Processing*, 4: 156-171, 1975.
- GOODCHILD, M. A spatial analytical perspective on geographical information systems. *International Journal of Geographical Information Systems*, 1 (4): 327-334, 1987.



- GUTING, R.H. An Introduction to Spatial Database Systems. *VLDB Journal*, 3(4), October 1994.
- HARTSHORNE, R. Propósitos e natureza da geografia. São Paulo, Editora HUCITEC, Editora da Universidade de São Paulo, 2. Ed., 1978.
- MAGUIRE, D.; DANGERRAMOND, J. The Functionality of GIS. In: Maguire, D.; Goodchild, M.; Rhind, D. (eds) *Geographical Information Systems: Principles and Applications*. New York, John Wiley and Sons, 1991, pp. 319-35.
- ROSS, J. L. S. Análise Empírica da Fragilidade dos Ambientes Naturais e Antropizados. *Revista do Dpto. de Geografia da FFLCH-USP*. N. 7. pp. 65-74. 1993.
- TOMLIN, D. *Geographic information systems and Cartographic Modeling*. Prentice Hall, New York, 1990.
- TRICART, J. *Ecodinâmica*. IBGE-SUPREN, (Recursos Naturais e Meio Ambiente). Rio de Janeiro. 1977.
- TRICART, J.; KIEWIETDEJONGE, C. *Ecogeography and Rural Management: A Contribution to the International Geosphere-Biosphere Programme*. Essex. Longman Scientific & Technical. 1992.

